MEASURING SPATIAL DEPENDENCE WITH SEMIVARIOGRAMS

R. A. OLEA

EDITOR'S REMARKS

This is the third volume in the *Series on Spatial Analysis*, and the second to be authored by Ing. Ricardo Olea, of ENAP, the national petroleum company of Chile. This publication, with OPTIMUM MAPPING TECHNIQUES USING REGIONALIZED VARIABLE THEORY (*SSA* #2), is the result of a joint research effort conducted by the Kansas Geological Survey and Empresa Nacional del Petróleo (ENAP). The methods described have been extensively used for research and petroleum exploration in both Kansas and Chile. This cooperative research between industry and governments on an international level has proved remarkably productive, and demonstrates that scientific cooperation can transcend national and cultural boundaries for the benefit of all.

MEASURING SPATIAL DEPENDENCE WITH SEMIVARIOGRAMS is unique among studies on structural analysis in that real data are used to illustrate the various forms which the semivariogram can assume. These results are representative of what will be obtained from actual regionalized variables, rather than idealized abstractions. This publication and the included computer program should, therefore, be especially valuable to those wishing to perform structural analysis and universal kriging for practical purposes.

Dr. John C. Davis, Editor
Series on Spatial Analysis

MEASURING SPATIAL DEPENDENCE WITH SEMIVARIOGRAMS

No. 3 in the Series on Spatial Analysis R.A. OLEA*/1977

ABSTRACT

The semivariogram is a measure of statistical dependence and is closely related to the autocorrelogram. It is particularly suitable for analysis of spatial functions that describe natural phenomena characterized by a geographic distribution and variations which are in part continuous and in part stochastic. If the phenomenon is sampled at regular intervals along lines, important statistical parameters can be estimated from the semivariogram.

INTRODUCTION

Spatial dependence may be expressed by a type of autocorrelogram referred to as the *semivariogram*, which expresses the degree of spatial variation as a function of distance. Semivariance is a special form of second-order moment used in the statistical theory of regionalized variables (Matheron, 1965).

Typically, a *regionalized variable* is a function that describes a natural phenomenon which has a geographic distribution. Examples of geologic properties which can be considered as regionalized variables include variations in structural elevation of a stratigraphic horizon (Olea, 1974), differences in Bouguer gravity values (Huijbregts and Matheron, 1971), changes in porosity within a petroleum reservoir, and changes in ore content in a mineralized body (Blais and Carlier, 1968).

A regionalized variable has the following characteristics: (1) An *observation*, which is a value from a function whose argument contains geographic coordinates specifying the location where the observation was made. Observations may be punctual, such as measurements of the gravitational or magnetic field taken at point locations, or they may consist of measurements made on areas or volumes, such as analyses of ore samples. If the observations are not punctual, the size, shape, and orientation of the samples must be specified. (2) An *average continuity*, in a mathematical sense. The spatial variation of a regionalized variable may be great or small, depending on the phenomenon, but continuity from point to point must exist. (3) A *random* or *stochastic component* which has no

CONTENTS

**EMPRESA NACIONAL DEL PETRÓLEO, Santiago, Chile.*

spatial continuity. Not all of the variation in the phenomenon may be expressable by a deterministic function.

Because regionalized variables include random components, they may be considered as random functions. Classical univariate statistics, however, is not adequate for the study of regionalized variables, because the information concerning the geographic locations of the samples cannot be considered (Blais and Carlier, 1968). The theory of regionalized variables as developed by Matheron (1965, 1970, 1971) takes into account the effects of sample location in the calculation of statistics. By utilizing the degree of relationship between adjacent samples, optimal interpolation and prediction can be achieved.

The semivariogram, or intrinsic function, is used in the theory of regionalized variables to express spatial variation. In form and use, it is similar to the autocorrelation function as used for signal analysis in communication theory (Mix, 1969). Let $Z(\vec{x} + \vec{h})$ and $Z(\vec{x})$ be values of the regionalized variable at two locations a distance $\vec{h}$ apart. The semivariogram is defined as:

$$\gamma(\vec{h}) = 1/2\ E[\{Z(\vec{x} + \vec{h}) - Z(\vec{x}) - E[Z(\vec{x} + \vec{h}) - Z(\vec{x})]\}^2]$$

The first and second moments of the difference $Z(\vec{x} + \vec{h}) - Z(\vec{h})$ depend only on the distance $\vec{h}$ between the two locations $(\vec{x} + \vec{h})$ and $\vec{x}$ and not on the individual coordinates. E is the expected value operator as used in classical statistics.

ESTIMATION OF THE SEMIVARIOGRAM

The expected value of a regionalized variable is called the *drift* (Matheron, 1969; Olea, 1975), and is conceptually similar to the "trend" as used in trend surface analysis. The drift accounts only for those slowly varying components of the regionalized variable which can be represented by analytical functions, at least within a local area or neighborhood. In this paper, the analytical functions used to represent the drift are polynomials of the zeroth, first, or second degree. Specifying that a polynomial drift is of zeroth degree is equivalent to stating that the regionalized variable is first-order stationary, because a polynomial of the zeroth degree is a constant.

The difference between the regionalized variable and its expected value or drift is called the *residual*, and is itself a regionalized variable:

$$R(\vec{x}) = Z(\vec{x}) - E[Z(\vec{x})]$$

The continuity of a regionalized variable may be expressed by a semivariogram of the residuals rather than in terms of the regionalized variable itself. If the drift is a constant, or of zeroth degree, the semivariogram of the regionalized variable and the semivariogram of the residuals will coincide.

If the residuals are second-order stationary, the semivariogram and the autocovariance of the residuals are related (Olea, 1975):

$$\gamma(\vec{h}) = Var[R(\vec{x})] - Cov(\vec{h})$$

This relationship is shown in Figure 1. The semivariogram shown was calculated for gravity residuals measured along a 3 km traverse in Toltén, Chile, shown in Figure 2. In the equation, $Var[R(\vec{x})]$ is the variance of the residuals and $Cov(\vec{h})$ is the autocovariance between residuals measured a distance $\vec{h}$ apart.

The most convenient way of estimating a semivariogram is from samples which are taken at regular intervals along a line. The sample locations can then be specified by a single coordinate and the rate of change in the regionalized variable is independent of the sampling pattern. Let $r(x_1)$, $r(x_2)$, $r(x_3)$,..., $r(x_t)$ be



FIGURE 1. Semivariogram of gravity residuals measured along a traverse near Toltén, south of Concepción, Chile. Horizontal scale gives distance (h) between successive pairs of gravity stations and is measured in kilometers. Left vertical scale is semivariance, right scale is autocovariance. Both are measured in milligals^2. Note that autocovariance scale is inverted because of reciprocal relation with semivariance for a stationary process. Light line represents variance of entire traverse, equal to 4.2 milligals^2.

residuals measured at regularly spaced points along a line. If the samples are taken from a single population having a semivariogram, it may be estimated by

$$\bar{\gamma}(pa) = \frac{1}{2(k-p)} \sum_{j=k'}^{k+k'-p-1} [r(x_j+pa) - r(x_j)]^2$$

for $k+k' < t$ and $p = 0, 1, \ldots, k-1$.

Here, a is the sampling interval and k is the length of the neighborhood, or interval over which the analytical expression for the drift used to generate the residual is valid. The leftmost sample in the interval is k', and in practice varies from 1 to t-k. The semivariogram for the sequence is taken as the average for all partial semivariograms. Although the semivariogram is a continuous function in concept, the fact that samples are taken at discrete intervals means that it can be estimated only at multiples of the sampling interval.

If a regionalized variable is not first-order stationary, the residuals can only be estimated, because they can be obtained only from an estimate of the drift. In such circumstances, the estimated semivariogram is biased by a predictable amount. When the drift is not a constant, semivariogram estimation must be done in an iterative fashion (Olea, 1975). To find the residuals which are necessary to estimate the semivariogram, the expression for the drift must be known. To calculate the drift expression, however, requires the semivariogram. This paradox can be resolved by a repeated estimation process.

The estimation process starts by assuming a semivariogram for the residuals and assuming an expression for the drift. The assumed semi-



FIGURE 2. Gravity residuals, in milligals, measured along traverse near Toltén, Chile. Spacing between stations is 500 meters.



FIGURE 3. Semivariogram of elevations of top of Cretaceous Springhill Formation measured along marine seismic traverse in Straits of Magellan, Chile. Semivariance is in meters². Light line represents sill, or variance of elevations, and is equal to 8.38 meters². Range, L, is distance beyond which difference between semivariance and sill is considered negligible and is equal to 3.5 km.

variogram may be any function which provides a reasonable fit to the real semivariogram. In practice, an estimate in the form of a sloping straight line provides a convenient initial assumption. Over short intervals of h, which are represented near the origin of the semivariogram, a straight line usually is a good approximation. The assumed expression for the drift may be a polynomial expansion of the geographic coordinates.

The second step in the recursive procedure is to estimate the coefficients in the polynomial expression used to approximate the drift. Estimated residuals are then found by subtracting the estimated drift from the observations. Finally, the estimated residuals are used to calculate an experimental semivariogram.

If all assumptions about the form of the semivariogram and the expression for the drift are correct, the experimental and assumed semivariograms should be identical. If the two semivariograms do not coincide, the process is repeated, with either a different assumed expression for the drift, a different semivariogram, or both.

If the assumed semivariogram is redefined, a smaller neighborhood is usually selected, because it is easier to represent the curve of the semivariogram by a line as the interval becomes shorter. The experimental semivariogram is independent of the slope of the assumed semivariogram. However, the slope does influence the correction necessary to remove the bias. Program SEMIVAR, listed in the Appendix, performs all necessary calculations for recursive estimation of experimental semivariograms.

USE OF THE SEMIVARIOGRAM

Parameters of the semivariogram must be known in order to apply regionalized variable techniques, such as mapping by universal kriging (Olea, 1974). For example, the kriging module of the SURFACE II graphics system (Sampson, 1975) requires specification of degree of polynomial drift, neighborhood size, and slope of the semivariogram as input parameters. However, the semivariogram also expresses characteristics of the regionalized variables which are important in their own right.

The slope of the semivariogram is an expression of the rate at which samples become increasingly independent with increasing distance. If the semivariogram asymptotically approaches an upper limit, there is a distance, called the *range*, which defines the minimum separation between statistically independent pairs of samples. The zone of influence of a sample extends over a distance equal to the range. Mathematically, the zone of influence is a sphere (or in two-dimensional mapping problems, a circle) whose radius is the smallest distance L such that:

$$\mathrm{Var}[R(\vec{x})] - \gamma(\vec{L}) = \varepsilon$$

where ε is a value so small that it may be considered negligible (Fig. 3). L is the range of the semivariogram. That part of the semivariogram to the right of the range is called the *sill* (Matheron, 1971). The semivariogram shown in Figure 3 was calculated from depth measurements of the top of the Cretaceous Springhill Formation along a seismic line in the Straits of Magellan (Olea and Davis, 1977). The subsurface structural measurements from which the semivariogram was calculated are shown in Figure 4.

In practice, the semivariogram may fluctuate around the sill as in Figure 1, rather than asymptotically approach an upper limit. In those situations, the range must be defined using different rules. For instance, the range may be considered to be either the distance to the leftmost maximum in the semivariogram or the distance to the first intersection of the sill and the semivariogram.

All observations within a distance to the point being estimated less than $\vec{L}$ provide information about the point. All samples outside the neighborhood defined by $\vec{L}$ are independent with respect to the point to be estimated, and may be disregarded because they do not provide any information about the regionalized variable at that location. The range is a fundamental parameter which must be estimated in order to achieve the optimum spacing of samples.

A regionalized variable is *anisotropic* if it possesses a "grain" or preferred orientation. Anisotropy is shown if the semivariogram varies in different directions, as in Figure 5. These semivariograms are calculated for bathymetric measurements along and across the continental shelf in front of Chiloé Island, Chile. The two bathymetric traverses are approximately 45 and

87 km in length, with bottom soundings taken at 1-km intervals (Fig. 6). Differences may be expressed in the slope at the origin, in the range, or in the sill. If the regionalized variable is isotropic, the semivariogram depends only on the magnitude of the vector $\vec{h}$ and not on its direction.

Figure 7 shows a special type of partial anisotropy called *geometric anisotropy*. In such instances, only the slope at the origin and the range are anisotropic; the sill remains the same regardless of orientation. The example is calculated for residual magnetic intensities measured by shipboard magnetometer along two traverses on the Pacific shelf near Concepción, Chile. The two traverses are shown in Figure 8.

If $\vec{h}$ is an n-dimensional vector, the semivariogram may not be a function of all n coordinates. For instance, in a three-dimensional layered body such as a stratiform ore deposit, the regionalized variable may change only with the third coordinate, depth. That is, the variable is constant on planes parallel to the horizontal geographic axes. If this is the case, the semivariogram will be a function only of the third component of $\vec{h}$. This is called *zonal anisotropy*, resulting from zonation of the regionalized variable.

The shape of the semivariogram, and in particular its behavior near the origin, provides information about the continuity and regularity of the regionalized variable. Figure 9 shows a semivariogram having a parabolic form near the origin. It represents behavior of the elevation of the erosional surface marking the top of the Cretaceous section in Río del Oro, Magallanes, Chile. The data, measured by seismic reflection, are shown in Figure 10. The shape of the



FIGURE 4. Subsurface structural elevations of top of Cretaceous Springhill Formation, estimated from reflection seismic measurements along 21-km marine traverse in Straits of Magellan. Depths in meters below sea level. Seismic measurements taken at 300-meter intervals.



FIGURE 5. Semivariograms of bathymetric measurements of sea bottom along (open circles) and across (solid dots) continental shelf in front of Chiloê Island, Chile. Intrinsic function is anisotropic, as semivariograms differ in slope, range, and sill.



FIGURE 6. Bathymetric traces parallel (open circles) and perpendicular (solid dots) to coast on continental shelf in front of Chiloê Island, Chile. A submarine canyon was crossed at 10 km on the traverse parallel to the coast. Bathymetric measurements were taken at 1-km intervals.

semivariogram implies excellent continuity in the regionalized variable, as $\gamma(\vec{h})$ is twice differentiable in $\vec{h} = 0$.

Figure 11 is linear near the origin, which implies moderate continuity, as $\gamma(\vec{h})$ is continuous but not twice differentiable. The semivariogram has been calculated for elevations of the top of the Jurassic Tobífera Series along a traverse in Magallanes, Chile. Figure 12 shows the seismic traverse from which the semivariogram was calculated.

The fact that $\gamma(0) = 0$ does not forbid $\lim \gamma(\vec{h}) = C,\ h \to 0$. In this situation, the semivariogram is discontinuous at the origin because of poor continuity of the regionalized variable. Such behavior is exhibited by placer deposits of gold, where the metal occurs as discrete, discontinuous particles, leading to the expression "nugget effect" for the constant, C. The example in Figure 13 is calculated from the permeability cores recovered from the Springhill Formation in the discovery well of the Poseśion gas field, Magallanes, Chile. The measurements, made on one-inch horizontal core plugs one inch in diameter, are given in Figure 14.

In Figure 15, the semivariogram is highly erratic and has no discernible pattern. Samples are not related to even their closest neighbors. Such a regionalized variable is a pure random function, and in this instance was generated from a sequence of 100 random numbers (Fig. 16). The same analytical results would be obtained by the use of either regionalized variable theory or by classical statistical methods which assume independent random variables.

Figure 17 shows a transitive form of semivariogram which is characterized by a finite range. In other words, it possesses a sill. This semivariogram indicates that there is moderate continuity within local neighborhoods, but the behavior of the phenomenon over longer



FIGURE 7. Semivariograms of residual magnetic intensities, showing geometric anisotropy. Magnetic intensities measured by shipboard magnetometer along two traverses on Pacific shelf near Concepción, Chile. Semivariogram represented by open circles has range (L_1) of 1.5 km; semivariogram represented by solid dots has range (L_2) of 4.5 km. Both have a sill of 170 gammas^2.

intervals is purely random. The example in Figure 17 represents topographic elevations measured along parallel 37°38'S, from the Atlantic shoreline to the Andes Mountains in Chile. The topographic profile is shown in Figure 18.

Regionalized variables may tend to repeat if examined over sufficiently large intervals. Under these circumstances dependence between samples does not decrease continuously with distance, as observations spaced far apart may be more similar than those which are closely spaced. The semivariogram of a cyclic regionalized variable is also cyclic in appearance. The frequency components of the semivariogram will be the average of those observed in the sampled sequence. Minima on a cyclic semi-



FIGURE 8. Residual magnetic intensities, measured along two perpendicular traverses by shipboard magnetometer over Pacific shelf near Concepción, Chile. Readings taken at 500-meter intervals. Residual intensities in gammas.

FIGURE 9. Semivariogram showing parabolic form near origin. Regionalized variable is structural elevation of top of Cretaceous System in Río del Oro, Magallanes, Chile. Elevations estimated from seismic reflections.



variogram occur at distances $\vec{h}$ which correspond to the wavelengths of pronounced repetitive elements in the data sequence. The relative importance of a cyclic component is expressed inversely by the magnitude of the semivariance at these minima. Figure 19 is the semivariogram for part of a gamma-ray log from an oil well in northwestern Kansas, drilled through a succession of limestones and shales of Pennsylvanian age. The semivariogram shows a minimum at $\vec{h}$ = 48 feet and a minor low at $\vec{h}$ = 18 feet. These reflect the tendency of these limestone units to repeat at approximately 48-foot intervals. The weaker 18-foot repeat seems to reflect the average thickness of lithologic units in the sequence. The gamma-ray log trace from which the data are taken is shown in Figure 20.

INFLUENCE OF SAMPLE SIZE

In general, reliability of a statistical study is proportional to the number of samples used. The larger the data set, the closer the statistics calculated from the set will estimate the true parameters. Figure 21 is a semivariogram for the regionalized variable used to calculate Figure 15, but using 5000 data points rather than 100 points. Data for both examples were sampled from a uniform distribution over the interval (0,1), having a variance of 0.08333. Because the data consist of a purely random sequence, the semivariogram should have the form of a horizontal line at $\gamma(h)$ = 0.08333. Comparison of Figures 15 and 21 shows that the true form of the semivariogram is more closely approximated with the larger sample size. In Figure 21, the scattering is almost negligible and the sill has moved to the correct location. This



FIGURE 10. Seismic profile showing estimated subsea elevations of top of Cretaceous System in Río del Oro, Magallanes, Chile. Seismic measurements spaced at 300-meter intervals.



FIGURE 11. Semivariogram showing linear form near origin. Regionalized variable is structural elevation of top of Jurassic Tobífera Series in Straits of Magellan, Chile. Elevations estimated from seismic reflections.



FIGURE 12. Marine seismic profile showing estimated subsea elevations of top of Jurassic Tobífera Series in Straits of Magellan, Chile. Seismic measurements spaced at 300-meter intervals.



FIGURE 13. Semivariogram showing nugget effect, or failure to go through origin. Regionalized variable is permeability of Springhill Formation in Posesión gas field, Magallanes, Chile. Semivariance in darcies², equal to 1 × 10⁶ millidarcies².



FIGURE 14. Permeability measured on one-inch horizontal core plugs taken at 1-meter intervals through the Springhill Formation. Core taken from discovery well in Posesión gas field, Magallanes, Chile. Permeability measured in millidarcies.



FIGURE 15. Semivariogram of random phenomenon having no spatial continuity. Data taken from uniform random distribution. Light line represents sill or variance equal to 0.093.



FIGURE 16. Sequence of 100 random numbers selected uniformly from the interval between zero and one.

example demonstrates that all possible information should be used in the calculation of semivariograms. A study should not be limited to a few lines if a larger data set is available. Longer sequences of points yield smoother semivariograms whose shapes are better approximations of the true or population semivariogram than are semivariograms obtained from smaller samples.

CONCLUSIONS

The semivariogram is a second-order moment of the differences between pairs of residuals of regionalized variables. Estimation of the semivariogram requires samples taken at regular intervals along lines.

The semivariogram can be used to numerically characterize the rate and amount of variation in regionalized variables. The characterizing statistics include the shape of the semivariogram near the origin, its slope, and the sill. A FORTRAN IV program for computing semivariograms from individual sequences or as averages of the semivariograms for a series of sequences is given in the Appendix.



FIGURE 17. Semivariogram showing transitive behavior. Regionalized variable is topographic elevation measured along Parallel 37°38'S, from Atlantic coast tc Andes Mountains in Chile.



FIGURE 18. Topographic elevations measured along Parallel 37°38'S from Atlantic coast on left to Andes Mountains on right. Measurements taken at 2-km intervals.



FIGURE 19. Semivariogram with cyclic behavior. Regionalized variable is gamma-ray intensity of alternating limestone-shale sequence of Pennsylvanian age, encountered in well in Rawlins Co., Kansas. Semivariance is expressed in API units^2. Cycles are evident at 48- and 18-foot intervals.



FIGURE 20. Digitized gamma-ray log from Bartasovsky No. 1 well in the Cahoj field, Rawlins Co., Kansas. Interval includes Pennsylvanian Lansing-Kansas City Group composed of alternating limestones and shales. Log digitized at 1-foot intervals.

REFERENCES

BLAIS, R.A. and CARLIER, P.A., 1968, Applications of geostatistics in ore evaluation: Canadian Institute of Mining and Metallurgy, Ore Reserve Estimation and Grade Control, Special Volume 9, Montreal, Quebec, p. 41-68.

HUIJBREGTS, C. and MATHERON, G., 1971, Universal kriging (an optimal method for estimating and contouring in trend surface analysis): Canadian Institute of Mining and Metallurgy, Decision-Making in the Mineral Industry, Special Volume 12, Montreal, Quebec, p. 159-169.

MATHERON, G., 1965, Les variables régionalisées et leur estimation. Une application de la théorie des fonctions aléatoires aux sciences de la nature: Masson et Cie., Editeurs, Paris, 305 p.

MATHERON, G., 1969, Le krigeage universal: Les Cahiers du Centre de Morphologie Mathématique de Fontainebleau, Fascicule 1, Edité par l'Ecole Nationale Supérieure des Mines de Paris, 83 p.

MATHERON, G., 1970, Random functions and their applications in geology, in Geostatistics, A Colloquium, D.F. Merriam, ed.: Plenum Press, New York, p. 79-87.

MATHERON, G., 1971, The theory of regionalized variables and its applications: Les Cahiers du Centre de Morphologie Mathématique de Fontainebleau, Fascicule 5, Edité par l'Ecole Nationale Supérieure des Mines de Paris, 211 p.

MIX, D.F., 1969, Random Signal Analysis: Addison-Wesley Publ. Co., Reading, Mass., 271 p.

OLEA, R.A., 1974, Optimal contour mapping using universal kriging: Jour. Geophysical Research, v. 79, no. 5, p. 695-702.

OLEA, R.A., 1975, Optimum mapping techniques using regionalized variable theory: Kansas Geol. Survey, Series on Spatial Analysis No. 2, Univ. Kansas, Lawrence, Kansas, 137 p.

OLEA, R.A. and DAVIS, J.C., 1977, Regionalized variables for evaluation of petroleum accumulation in Magellan Basin, South America: Am. Assoc. Petroleum Geologists Bull., v. 61, no. 4, in press.

SAMPSON, R.J., 1975, SURFACE II Graphics System: Kansas Geol. Survey, Series on Spatial Analysis No. 1, Univ. Kansas, Lawrence, Kansas, 240 p.



FIGURE 21. Semivariogram of random phenomenon having no spatial continuity. Data consist of 5000 random numbers selected uniformly from the interval between zero and one. Light line represents sill or variance of 0.083. Compare with Figure 15.

APPENDIX

SEMIVAR, A FORTRAN IV PROGRAM FOR SEMIVARIOGRAM ESTIMATION

This computer program calculates semivariograms for regionalized variables or their estimated residuals. If desired, an average semivariogram may be calculated for groups of parallel traverses, from the semivariograms of the individual lines. The semivariograms may be presented in either biased or unbiased form. Statistical comparisons can be made between the experimental semivariogram and the assumed semivariogram when applicable.

Program SEMIVAR incorporates several assumptions about the data and is subject to certain restrictions. Data are presumed to consist of sequences of observations taken at regularly spaced intervals along straight lines. If a semivariogram for the residuals must be assumed, the program uses a semivariogram having the form of a straight line through the origin, whose slope is derived from the slope of the experimental semivariogram at its origin. The analytical expression used to represent the drift is a polynomial whose maximum degree is two.

In one run, program SEMIVAR will calculate the semivariogram for samples collected along a single line; the average semivariogram for a group of parallel lines; the semivariograms for a series of single lines at different orientations; or the average semivariograms for groups of parallel lines at different orientations. A line is separated from other lines in a group by an end-of-sequence card. Groups are separated by two end-of-sequence cards in succession. In addition, each group begins with a group control card. A typical deck representing two groups each containing two lines might appear:

FORMAT CONTROL CARD

GROUP CONTROL CARD FOR GROUP 1

DATA CARDS FOR LINE 1

END-OF-SEQUENCE CARD

(cont.)

DATA CARDS FOR LINE 2

END-OF-SEQUENCE CARD

END-OF-SEQUENCE CARD

GROUP CONTROL CARD FOR GROUP 2

DATA CARDS FOR LINE 3

END-OF-SEQUENCE CARD

DATA CARDS FOR LINE 4

END-OF-SEQUENCE CARD

END-OF-SEQUENCE CARD

SYSTEM ENDJOB CARD

The first card of the data deck *must* be the FORMAT CONTROL CARD. All line data used by the program must be coded using the format specified. Each data card contains a single measurement of the variable (Z-value) whose semivariance is to be calculated; and an identification number or name for that point. The FORMAT CONTROL CARD specifies the format for these data cards:

Columns 1 - 40 Format of data cards, as a FORTRAN format specification enclosed in parentheses. The field for the Z-values must be specified as a real variable. The identifier (number or name) must be read in as a three-element alphanumeric array such as 3A4. The data are printed for checking purposes using F10.3 and 3A4 format specifications.

Column 50 If 1, the first field on a data card contains the Z-value.
If 2, the first field on a data card contains the identifier.

The GROUP CONTROL CARD specifies the parameters to be used in calculating the semivariogram of the data which follows. Specifications for this card are:

Columns	Format	Variable Name
1-5	I5	N
6-10	I5	ID
11-15	I5	NPLL
16-20	I5	IB
21-25	F5.0	DL
26-30	A4,A1	UNIT(2)
31-78	12A4	TIT(12)

Note that all integer variables (N, ID, NPLL, and IB) must be right-justified within the proper field.

N	Number of sample points within the neighborhood.
ID	Desired polynomial degree of the drift, which may be 0, 1, or 2.
NPLL	If 0, the semivariogram for each line in the group will be printed, as well as the average semivariogram. If 1, only the average semivariogram will be printed.
IB	If 0, only the biased form of the semivariogram will be printed. If 1, both biased and unbiased forms will be printed.
DL	Distance between successive sample points.
UNIT	Distance measurement units (FEET, METERS, MILES, etc.)
TIT	Alphanumeric information to be used as titles.

DATA CARDS contain Z-values on which the semivariogram is to be calculated, plus an identification number or name containing up to 12 alphanumeric characters. The format of the data cards must correspond to the format specification in the FORMAT CONTROL CARD.

END-OF-SEQUENCE CARDS are used to denote the end of a line and the end of a group of lines. They consist of the number 9999. punched in the field containing the Z-value. If the data have a range which exceeds the EOS number, it can be changed by altering the DATA statement

```
DATA EOD/9998.0/
```

which immediately follows the DIMENSION statements.

```
C                    STRUCTURAL ANALYSIS USING SEMIVARIOGRAMS .
C
C***********************************************************************
C
C         AUTHOR : RICARDO A. OLEA
C                  EMPRESA NACIONAL DEL PETROLEO
C                  CASILLA 3556
C                  SANTIAGO , CHILE .
C
C         DATE : DECEMBER 1974
C
C         PROGRAMMING LANGUAGE : HONEYWELL 635 FORTRAN IV LANGUAGE
C
C         DIRECT INQUIRIES : AUTHOR , OR
C                            JOHN C. DAVIS
C                            GEOLOGIC RESEARCH SECTION
C                            KANSAS GEOLOGICAL SURVEY
C                            1930 AVENUE "A" , CAMPUS WEST
C                            THE UNIVERSITY OF KANSAS
C                            LAWRENCE , KANSAS 66044
C                                  913-864-4991
C
C         MEMORY REQUIRED : 12K .
C
C         PURPOSE:
C                 THIS IS A PROGRAM TO CALCULATE SEMIVARIOGRAMS FOR A
C         REGIONALIZED VARIABLE SAMPLED AT REGULAR INTERVALS ALONG A
C         LINE . THE DRIFT CAN BE A POLYNOMIAL OF DEGREE 0 , 1 OR 2 .
C
C         METHOD :
C                 LET M BE THE NUMBER OF SAMPLES ALONG A LINE . SUPPOSE
C         YOU ARE INTERESTED IN A SEMIVARIOGRAM FOR AN INTERVAL LENGTH
C         OF N SAMPLES , N NOT GREATER THAN M . THE PROGRAM SLIDES A
C         WINDOW N SAMPLES LONG OVER THE LINE . THE SEMIVARIOGRAM
C         FOR THE LINE IS THE AVERAGE OF ALL PARTIAL SEMIVARIOGRAMS .
C
C         THE N SAMPLE INTERVAL SEMIVARIOGRAM FOR A GROUP OF LINES
C         ALONG THE SAME DIRECTION IS THE AVERAGE OF ALL PARTIAL
C         SEMIVARIOGRAMS .
C
C         DATA :
C               THERE IS ONE CARD PER SAMPLE POINT , TWO FIELDS PER
C         CARD . THE ESSENTIAL INFORMATION IS THE REGIONALIZED VARIABLE
C         VALUE . THERE IS AN OPTIONAL FIELD 12 ALPHANUMERIC CHARACTER
C         LONG FOR IDENTIFICATION PURPOSES . THE FORMAT AND RELATIVE
C         LOCATION IS OPTIONAL . HOWEVER , THE FORMAT MUST BE ABLE
C         TO READ A REAL VARIABLE AND ALPHANUMERIC INFORMATION
C         FOR ANOTHER THREE VARIABLES .
C
C         DATA STRUCTURE IS :
C
C
C         VARIABLE FORMAT CARD
C         CONTROL CARD FOR THE FIRST GROUP OF LINES
C         DATA FOR LINE 1
C         ENDSEQ
C         DATA FOR LINE 2
C         ENDSEQ
C               .
C               .
C               .
C         DATA FOR THE LAST LINE IN GROUP 1
C         ENDSEQ
C         ENDSEQ
```

```
C          CONTROL CARD FOR THE SECOND GROUP OF LINES
C               .
C               .
C               .
C          DATA FOR THE LAST LINE IN THE LAST GROUP
C          ENDSEQ
C          ENDSEQ
C          $        ENDJOB
C          THE ENDSEQ MUST GO IN THE REGIONALIZED VARIABLE VALUE FIELD
C         AND  MUST BE ANY NUMBER LARGER THAN VARIABLE EOD SPECIFIED IN
C         THE PROGRAM .
C
C          THE VARIABLE FORMAT CARD TO READ IN DATA HAS THE FOLLOWING
C          FIELDS:
C
C          COLUMN  1-40 : VARIABLE FORMAT TO READ IN DATA (10A4)
C          COLUMN    50 : THE INTERGER VALUE 1 OR 2 TO DEFINE RELATIVE
C                         LOCATION BETWEEN THE ID FIELD AND THE
C                         REGIONALIZED VARIABLE VALUE FIELD . 1 MEANS
C                         THE REGIONALIZED VARIABLE COMES BEFORE THE
C                         IDENTIFICATION AND 2  OTHER WAY AROUND. THE
C                         THE ASSIGNED VALUE IS 1 .
C
C          THE CONTROL CARD FORMAT IS :
C
C          COLUMN  1- 5 : NUMBER OF SAMPLE POINTS IN THE WINDOW (I5).
C                  6-10 : THE POLYNOMIAL DEGREE FOR THE DRIFT (I5) .
C                 11-15 : PRINTING OPTION (I5) . A ZERO WILL PRINT
C                         THE SEMIVARIOGRAM FOR ALL SEQUENCES PLUS
C                         THE AVERAGE SEMIVARIOGRAM . A ONE WILL
C                         ONLY PRINT THE AVERAGE SEMIVARIOGRAM FOR
C                         THE GROUP . THE ASSUMED VALUE IS 0 .
C                 16-20 : FOR THE CASE THE POLYNOMIAL DEGREE FOR
C                         THE DRIFT IS  1 OR 2 , A ZERO WILL
C                         PRINT ONLY THE BIASED AVERAGE SEMIVARIOGRAM.
C                         A ONE WILL PRINT BOTH THE BIASED AND
C                         THE UNBIASED AVERAGE SEMIVARIOGRAMS .
C                 21-25 : DISTANCE BETWEEN SUCCESSIVE SAMPLES (F5.0) .
C                 26-30 : UNIT USED IN MEASURING THE DISTANCE (A4,A1) .
C                 31-78 : COMMENTS (12A4) .
C
C          SPECIAL SUBROUTINES:
C                              THE PROGRAM CALLS A SUBROUTINE TO PLOT
C          THE SEMIVARIOGRAMS . IN CURRENT VERSION THAT SUBROUTINE NAME
C          IS  M15A .
C
C          RESULTS :
C                   IF DESIRED , THE PROGRAMS OFFERS THE POSSIBILITY
C          TO LIST THE INPUT DATA , PLOT THE SEMIVARIOGRAMS AND
C          PRINT A SEMIVARIOGRAM TABLE FOR EACH
C          DATA SEQUENCE . FOR THE GROUP OF LINES THERE IS A
C          TABLE FOR THE AVERAGE SEMIVARIOGRAM AND A GRAPHIC DISPLAY .
C          BOTH IN THE TABLE AND IN THE GRAPH THERE IS A COMPARISON TO
C          AN ASSUMED SEMIVARIOGRAM TO DETERMINE THE GOODNESS OF
C          FIT OF THE SEMIVARIOGRAM AND DRIFT CHOICE TO REALITY .
C          THE COMPARISON CAN BE DONE IN TERMS OF THE BIASED
C          SEMIVARIOGRAMS ONLY OR FOR BOTH BIASED AND UNBIASED
C          SEMIVARIOGRAMS .
C
C***********************************************************************
C
      DIMENSION Z(300),GAM(300),VAR(300),Y(300),FV(300),FT(10)
      DIMENSION X(300),TV(300),H(300),TIT(12),POINT(3,300),UNIT(2)
      DATA IFILE/11/,JFILE/6/
```

```
      DATA EOD/998.0/
C              DATA I/O
C
      READ(IFILE,95)FT,IOR
      IF(IOR .LE. 0 .OR. IOR .GT. 2) IOR = 1
   42 READ(IFILE,96,END=47)N,ID,NPLL,IB,DL,UNIT,TIT
      ID =ID + 1
      KOUN = 0
      DO 21 K = 1,N
      X(K) = K - 1
   21 FV(K) = 0.
      TS = 0.
   13 M = 0
      GO TO(69,70),IOR
   69 DO 71 I = 1,301
      READ(IFILE,FT,ERR=50) Z(I),(POINT(II,I),II=1,3)
      IF(Z(I) .GE. EOD) GO TO 50
   71 M = M + 1
      GO TO 73
   70 DO 72 I = 1,301
      READ(IFILE,FT,END=50) (POINT(II,I),II=1,3),Z(I)
      IF(Z(I) .GE. EOD) GO TO 50
   72 M = M + 1
   73 WRITE(JFILE,89)
      STOP
   50 IF(M)23,23,11
   11 KOUN = KOUN + 1
      IF(NPLL .GT. 0) GO TO 33
      WRITE(JFILE,92)TIT,DL,UNIT,KOUN
      DO 24 J = 1,M
   24 WRITE(JFILE,103)J,(POINT(I,J),I=1,3),Z(J)
      GO TO(60,31,32),ID
   60 SUMM = 0.0
      SUMS = 0.0
      DO 80 J = 1,M
      SUMM = SUMM + Z(J)
   80 SUMS = SUMS + Z(J)*Z(J)
      SUMM = SUMM/M
      SUMS = SUMS/M - SUMM*SUMM
      WRITE(JFILE,111)N,SUMM,SUMS
      GO TO 33
   31 WRITE(JFILE,105)N
      GO TO 33
   32 WRITE(JFILE,106)N
   33 IF(M .GE. N) GO TO 12
      WRITE(JFILE,91)
      N = M
C
C              WINDOW ORIGIN
C
   12 INT = M - N + 1
      RM = M
      RN = N
      N1 = N - 1
      RN1 = N1
      DO 19 K = 1,N1
   19 VAR(K) = 0.
      DO 14 I = 1,INT
      NI = N  + I - 1
      RI = I
      GO TO(61,34,35),ID
C
```

```
C                RESIDUALS FOR A STATIONARY DRIFT .
C
   61 DO 62 K = 1,N
      KI = K + I
   62 Y(K) = Z(KI - 1)
      GO TO 38
C
C                FIRST DEGREE POLYNOMIAL COEFFICIENT .
C
   34 A1 = (Z(NI) - Z(I))/RN1
C
C                RESIDUALS FOR FIRST DEGREE DRIFT .
C
      DO 37 K = 1,N
      KI = K + I
      RK1 = K - 1
   37 Y(K) = Z(KI - 1) - A1*RK1
      GO TO 38
C
C                SECOND DEGREE POLYNOMIAL COEFFICIENTS .
C
   35 ZM = 0.
      DO 15 JJ = I,NI
   15 ZM = Z(JJ) + ZM
      ZM = 2.*ZM/RN
      A2 = (3.*(Z(NI) + Z(I) - ZM))/((RN1 - 1.0)*RN1)
      A1 = (Z(NI) - Z(I))/RN1 - RN1*A2
C
C                RESIDUALS FOR SECOND DEGREE DRIFT .
C
      DO 16 K = 1,N
      KI = K+ I
      RK1 = K - 1
   16 Y(K) = Z(KI - 1) - A1*RK1 - A2*RK1*RK1
C
C                SEMIVARIOGRAM FOR A WINDOW .
C
   38 DO 17 K = 1,N1
      GAM(K) = 0.
      IMAX = N - K
      RMAX = N - K
      DO 18 L = 1,IMAX
      LK = L + K
   18 GAM(K) = GAM(K) + (Y(L) - Y(LK))*(Y(L) - Y(LK))
   17 GAM(K) = GAM(K)/( 2.0*RMAX)
C
C                SEMIVARIOGRAM FOR A LINE .
C
      DO 14 K = 1,N1
   14 VAR(K) = VAR(K) + GAM(K)
      DO 20 KK = 1,N1
      K = N - KK + 1
   20 VAR(K) = VAR(K-1)/RI
      VAR(1) = 0.
      IF(NPLL .GT. 0) GO TO 49
      CALL M15A(1,X,VAR,N, , , ,0)
      WRITE(JFILE,104)TIT,M,N,DL,UNIT
      GO TO (63,51,52),ID
   63 WRITE(JFILE,112)
      GO TO 53
   51 WRITE(JFILE,100)
      GO TO 53
   52 WRITE (JFILE,101)
```

```
   53 WRITE(JFILE,98)
      DO 48 K = 1,N
      KL = K - 1
   48 WRITE(JFILE,97)KL,VAR(K)
C
C                  SEMIVARIOGRAM FOR A GROUP OF LINES .
C
   49 FV(1) = 0.
      DO 22 K = 2,N
   22 FV(K) = FV(K) + VAR(K)*RI
      TS = TS + RI
      GO TO 13
   23 DO 25 K = 2,N
   25 FV(K) = FV(K)/TS
C
C                  ASSUMED SEMIVARIOGRAM .
C
      GO TO(64,45,44),ID
C                  LINEAR DRIFT
   45 CF = RN1*FV(2)/(RN1 - 1.0)
      GO TO 43
C                  QUADRATIC DRIFT .
   44 AUX1 = RN1*(RN1*RN1 - 1.0)
      AUX2 = 2.0*RN1*(RN1 + 1.0) - 1.0
      CF = RN1*FV(2)/(RN1 - 2.0)
   43 DO 26 I = 1,N
      H(I) = I - 1
      GO TO (47,28,29),ID
C                  LINEAR DRIFT
   28 TV(I) = CF*H(I)*(1.0 - H(I)/RN1)
      GO TO 26
C                  QUADRATIC DRIFT .
   29 AUX = AUX2 - 2.0*H(I)*(RN1 + 1.0) + H(I)*H(I)
      TV(I) = CF*H(I)*(1.0 - H(I)*AUX/AUX1)
   26 CONTINUE
C
C                  RESULT PRINTING .
C
   64  GO TO(65,66,66),ID
   65 CALL M15A(1,X,FV,N, , , ,0)
      WRITE(JFILE,113)TIT,KOUN,N,DL,UNIT
      DO 67 K = 1,N
      KL = K - 1
   67 WRITE(JFILE,97)KL,FV(K)
      CF = FV(2)
      GO TO 74
C           BIASED SEMIVARIOGRAMS .
   66 IBI = 0
   78 CALL M15A(2,X,TV,N,X,FV,N,0)
      WRITE(JFILE,93)TIT,KOUN,N,DL,UNIT
      GO TO (74,54,55),ID
   54 WRITE(JFILE,100)
      GO TO 56
   55 WRITE(JFILE,101)
   56 IF( IBI .LE. 0) GO TO 75
      WRITE(JFILE,115)
      IB = 0
      GO TO 76
   75 WRITE(JFILE,114)
   76 WRITE(JFILE,109)
      DO 39 K = 1,N
      J = K - 1
   39 WRITE(JFILE,110)J,FV(K),J,TV(K)
      IF(IB .LE. 0) GO TO 74
```

```
C        UNBIASED SEMIVARIOGRAMS .
      DO 77 I = 1,N
      RI = I -1
      FV(I) = RI*CF + FV(I) - TV(I)
   77 TV(I) = RI*CF
      IBI = 1
      GO TO 78
   74 CF = CF/DL
      WRITE(JFILE,108)CF,UNIT
   89 FORMAT(//' END OF DATA WAS NOT FOUND UPON READING 300 SAMPLES')
   91 FORMAT(///' THE WINDOW IS LONGER THAN THE LINE .')
   92 FORMAT(1H1,1X,12A4//' SAMPLE DISTANCE',F10.2,2X,2A4//
     1      ' THIS IS LINE'    ,1I3,' FOR THE GROUP'////)
   93 FORMAT(/' EXPERIMENTAL SEMIVARIOGRAM (X) AND ASSUMED (*)',
     1   23X,12A4/' NUMBER OF LINES FOR THE GROUP : ',I2,
     2   5X,'NUMBER OF SAMPLES IN THE WINDOW : ',I3,
     3        5X,'SAMPLE DISTANCE',F10.2,2X,2A4)
   95 FORMAT(10A4,5X,1I5)
   96 FORMAT(4I5,F5.0,1A4,1A1,12A4)
   97 FORMAT(32X,' GAMMA(',I2,') = ',F10.2)
   98 FORMAT(////29X,' EXPERIMENTAL SEMIVARIOGRAM'///)
  100 FORMAT(17X,'FIRST DEGREE POLYNOMIAL FOR THE DRIFT')
  101 FORMAT(17X,'SECOND DEGREE POLYNOMIAL FOR THE DRIFT')
  103 FORMAT(' Z(',I3,')',5X,3A4,5X,F10.3)
  104 FORMAT(/' EXPERIMENTAL SEMIVARIOGRAM',40X,12A4/
     1  ' NUMBER OF SAMPLES IN THE LINE : ',I3,
     2  5X,' NUMBER OF SAMPLES IN THE WINDOW : ',I3,
     3       5X,' SAMPLE DISTANCE ',F10.2,2X,2A4)
  105 FORMAT(////' NUMBER OF SAMPLES IN THE WINDOW : ',I2///
     1      ' THE DRIFT IS A FIRST DEGREE POLYNOMIAL')
  106 FORMAT(////' NUMBER OF SAMPLES IN THE WINDOW : ',I2///
     1      ' THE DRIFT IS A SECOND DEGREE POLYNOMIAL ')
  108 FORMAT(///' SEMIVARIOGRAM SLOPE AT THE ORIGIN',4X,F10.4
     1      ,' SQUARE UNITS / ',A4,A1////)
  109 FORMAT(/////29X,' EXPERIMENTAL SEMIVARIOGRAM' ,T70,
     1      'ASSUMED SEMIVARIOGRAM'///)
  110 FORMAT(33X,' GAMMA(',I2,') = ',F10.4,T70,' GAMMA(',I2,') =',
     1      F10.4)
  111 FORMAT(////' NUMBER OF POINTS IN THE WINDOW : ',I2///
     1      ' THE DRIFT IS A CONSTANT'///
     2      ' THE MEAN IS ',F15.5,' AND THE VARIANCE ',F15.5)
  112 FORMAT(/47X,'THE DRIFT IS A CONSTANT'//)
  113 FORMAT(/' AVERAGE EXPERIMENTAL SEMIVARIOGRAM',33X,12A4/
     1          'NUMBER OF LINES USED : ',I2,
     2           5X,' NUMBER OF SAMPLES IN THE WINDOW : ',I3
     3          ,5X,' SAMPLE DISTANCE',F10.2,2X,2A4/47X,
     4          ' THE DRIFT IS A CONSTANT'//////25X,
     5          ' AVERAGE EXPERIMENTAL SEMIVARIOGRAM'///)
  114 FORMAT('+',80X,' BIASED SEMIVARIOGRAMS ')
  115 FORMAT('+',80X,' UNBIASED SEMIVARIOGRAMS')
      GO TO 42
   47 STOP
      END
```

TEST DATA

The following control cards were used to create a semivariogram of estimated depths from seismic return times along a marine seismic traverse in the Straits of Magellan. The first variable in the data set is estimated depth to the reflecting horizon, in meters, and the second variable is an identifier. The complete data set is not given, as it is listed as part of the program output. Note that on the output, the identifier is the first variable.

```
(F6.0,5X,3A4)                                          1
   61     0      0      0 300.METER    SEMIVARIOGRAM TEST DATA SET
 1882.       36 420
 1878.       36 410
 1867.       36 400
          .
          .
          .
          .
 1592.       36 758
 9999.
 9999.
```

EXAMPLE OUTPUT

```
          SEMIVARIOGRAM TEST DATA SET

       SAMPLE DISTANCE     300.00  METER

       THIS IS LINE   1 FOR THE GROUP


       Z(   1)      36  420               1882.000
       Z(   2)      36  410               1878.000
       Z(   3)      36  400               1867.000
       Z(   4)      36  390               1843.000
       Z(   5)      36  380               1831.000
       Z(   6)      36  370               1816.000
       Z(   7)      36  360               1802.000
       Z(   8)      36  350               1799.000
       Z(   9)      36  330               1757.000
       Z(  10)      36  320               1738.000
       Z(  11)      36  310               1729.000
       Z(  12)      36  300               1712.000
       Z(  13)      36  290               1684.000
       Z(  14)      36  280               1678.000
       Z(  15)      36  270               1660.000
```

Z(16)	36 260	1655.000
Z(17)	36 250	1634.000
Z(18)	36 230	1725.000
Z(19)	36 220	1771.000
Z(20)	36 210	1808.000
Z(21)	36 200	1834.000
Z(22)	36 190	1852.000
Z(23)	36 180	1864.000
Z(24)	36 170	1864.000
Z(25)	36 160	1871.000
Z(26)	36 150	1885.000
Z(27)	36 140	1881.000
Z(28)	36 130	1881.000
Z(29)	36 120	1872.000
Z(30)	36 110	1858.000
Z(31)	36 100	1831.000
Z(32)	36 90	1795.000
Z(33)	36 80	1776.000
Z(34)	36 70	1779.000
Z(35)	36 60	1769.000
Z(36)	36 50	1764.000
Z(37)	36 40	1730.000
Z(38)	36 30	1708.000
Z(39)	36 20	1696.000
Z(40)	36 10	1675.000
Z(41)	36 1	1692.000
Z(42)	36 450	1684.000
Z(43)	36 460	1692.000
Z(44)	36 470	1690.000
Z(45)	36 480	1673.000
Z(46)	36 500	1658.000
Z(47)	36 510	1661.000
Z(48)	36 520	1696.000
Z(49)	36 530	1730.000
Z(50)	36 540	1764.000
Z(51)	36 550	1751.000
Z(52)	36 560	1747.000
Z(53)	36 570	1737.000
Z(54)	36 580	1730.000
Z(55)	36 590	1718.000
Z(56)	36 600	1703.000
Z(57)	36 610	1683.000
Z(58)	36 620	1662.000
Z(59)	36 630	1629.000
Z(60)	36 640	1624.000
Z(61)	36 650	1613.000
Z(62)	36 660	1609.000
Z(63)	36 670	1613.000
Z(64)	36 680	1595.000
Z(65)	36 690	1586.000
Z(66)	36 700	1590.000
Z(67)	36 710	1600.000
Z(68)	36 720	1591.000
Z(69)	36 730	1562.000
Z(70)	36 758	1592.000

NUMBER OF POINTS IN THE WINDOW # 61

THE DRIFT IS A CONSTANT

THE MEAN IS 1733.27142 AND THE VARIANCE 8383.95422



EXPERIMENTAL SEMIVARIOGRAM SEMIVARIOGRAM TEST DATA SET
NUMBER OF SAMPLES IN THE LINE # 70 NUMBER OF SAMPLES IN THE WINDOW # 61 SAMPLE DISTANCE 300.00 **METER**

THE DRIFT IS A CONSTANT

EXPERIMENTAL SEMIVARIOGRAM

GAMMA(0)	=	0.
GAMMA(1)	=	252.46
GAMMA(2)	=	767.67
GAMMA(3)	=	1489.57
GAMMA(4)	=	2305.05
GAMMA(5)	=	3181.75
GAMMA(6)	=	4081.35
GAMMA(7)	=	4974.74
GAMMA(8)	=	5866.13
GAMMA(9)	=	6711.81
GAMMA(10)	=	7379.40
GAMMA(11)	=	7861.79
GAMMA(12)	=	8115.07

GAMMA(13)	=	8161.30
GAMMA(14)	=	8062.18
GAMMA(15)	=	7863.15
GAMMA(16)	=	7626.43
GAMMA(17)	=	7363.46
GAMMA(18)	=	7054.51
GAMMA(19)	=	6700.92
GAMMA(20)	=	6300.32
GAMMA(21)	=	5934.51
GAMMA(22)	=	5659.14
GAMMA(23)	=	5523.57
GAMMA(24)	=	5567.91
GAMMA(25)	=	5725.35
GAMMA(26)	=	5998.60
GAMMA(27)	=	6370.83
GAMMA(28)	=	6869.72
GAMMA(29)	=	7518.05
GAMMA(30)	=	8356.41
GAMMA(31)	=	9378.52
GAMMA(32)	=	10488.45
GAMMA(33)	=	11591.27
GAMMA(34)	=	12519.30
GAMMA(35)	=	13264.23
GAMMA(36)	=	13752.74
GAMMA(37)	=	13985.55
GAMMA(38)	=	13906.32
GAMMA(39)	=	13482.11
GAMMA(40)	=	12680.02
GAMMA(41)	=	11591.80
GAMMA(42)	=	10270.44
GAMMA(43)	=	8981.10
GAMMA(44)	=	7716.41
GAMMA(45)	=	6631.57
GAMMA(46)	=	5806.03
GAMMA(47)	=	5265.71
GAMMA(48)	=	5143.72
GAMMA(49)	=	5496.03
GAMMA(50)	=	6297.71
GAMMA(51)	=	7451.32
GAMMA(52)	=	8928.38
GAMMA(53)	=	10822.55
GAMMA(54)	=	12915.90
GAMMA(55)	=	15085.59
GAMMA(56)	=	17366.15
GAMMA(57)	=	19679.01
GAMMA(58)	=	21998.58
GAMMA(59)	=	24201.43
GAMMA(60)	=	26262.60



NUMBER THREE
SERIES ON SPATIAL ANALYSIS

KANSAS GEOLOGICAL SURVEY
1930 AVENUE A   CAMPUS WEST
LAWRENCE, KANSAS   U.S.A.   66044